
Code recap
' ░▒▓█'[⊃⌈4×9÷⍨+/2>|F\9⍴⊂¯2J¯1.5+3×⊖⍉a∘.+0J1×a←n÷⍨⍳1+n←10]

Grid ← {

(centre (w h)) ← ⍺ ⋄ re ← ⍵÷⍨⍳1+⌊w×⍵ ⋄ im ← ⌽0J1×⍵÷⍨⍳1+⌊h×⍵

centre+(im∘.+re)-(w÷2)+0J1×h÷2

}

Code recap
∇ its ← maxiter Mandelbrot grid

;m ;_ ;cs ;_cs ;F

_cs ← cs ← ,grid
m ← 1⍴⍨⍴cs
its ← 0⍴⍨⍴cs
F ← {⍺+⍵*2}
:For _ :In ⍳maxiter

(m/cs) ← _cs F⍥(m∘/) cs
(m/its) +← 1
(m/m) ← 2>|m/cs

:EndFor
its ⍴⍨← ⍴grid

∇

Code recap
Palette ← {

⍝ Dumb-down of https://stackoverflow.com/a/25816111/2828287

n ← ⌈⍵÷4

Seg ← {⌊⍺+⍤0 1⊢(⍵-⍺)∘.×n÷⍨⍳n}

0,⍨⊃,/2Seg/(0 7 100)(66 107 203)(237 255 255)(255 170 0)(0 2 0)

}

GreyPalette ← {

colours ← ⌊255×2*⍨⍵÷⍨⍳1+⍵

1⌽colours⍴⍨3,⍴colours

}

Code recap (usage)
its ← 50 Mandelbrot ¯.5 (3 2) Grid 300

pal ← GreyPalette 50

colours ← 256⊥pal[;its]

'c:/tmp/mdb.png' Png.Write colours

Useful links
• Jupyter notebook on the Mandelbrot set

https://github.com/Dyalog/dyalog-jupyter-notebooks
• Brian Becker’s post on Mandelbrot + isolates 

https://www.dyalog.com/blog/2014/08/isolated-mandelbrot-set-explorer/
• Nicolas Delcros’s “Making images and videos of the Buddhabrot” article in Vector

https://vector.org.uk/making-images-and-videos-of-the-buddhabrot/
o (Png namespace)

• Today’s code as a namespace
https://github.com/RojerGS/fractals/tree/master/mapl

https://github.com/Dyalog/dyalog-jupyter-notebooks
https://www.dyalog.com/blog/2014/08/isolated-mandelbrot-set-explorer/
https://vector.org.uk/making-images-and-videos-of-the-buddhabrot/
https://github.com/RojerGS/fractals/tree/master/mapl


5

dyalog.tv / @dyalogapl / #dyalog

Upcoming webinars
May 20 BAA open session
May 27 - - -
Jun 3 BAA open session
Jun 10 Dyalog Error Handling – Part 3
Jun 17 BAA AGM
Apr 29 - - -
July 1 BAA open session
July 8 Dyalog TBD

britishaplassociation.org

Thanks for watching!

Questions?


